



---

# プログラミング演習2 ネットワーク通信

---

中村, 高橋, 小林, 橋本

# 使ってますよね？




- LINE
- Twitter
- Facebook
- Skype
- FaceTime




# 本日の内容



- ネットワーク通信で情報のやりとりをしよう




# 通信の基本



ポートは 0 ~ 65535 までの  
番号を指定することが可能  
ポートは入り口の事

# 窓口とサービスの対応関係

明治大学総合数理学部  
先端メディアサイエンス学科  
中村研究室



# 市役所で考えてみる



1. 役所の係の人が窓口を開けて準備する
2. 市民が役所（の住所）に行き，窓口に並ぶ
3. 窓口の人が最初に並んでいる市民を呼ぶ
4. 窓口で，役所の人と，市民が対話して何らかの処理を行う



# サーバ・クライアント



1. サーバがポートを開けて準備する
2. クライアントがサーバのアドレスとポート番号を指定してサーバに接続要求を行う
3. サーバがクライアントに対して接続許可を出すことで、情報交換のための経路が確立される
4. サーバとクライアント間でやりとりが行われる

# 注意点



- ファイル名やフォルダ名にServerやClientという文字列を含まない事
- 下記をプログラムに入れること

```
void stop(){  
    myClient.stop();  
}
```

```
void stop(){  
    myServer.stop();  
}
```

- draw() が無いと動作しません
- 問題が解決しない場合は再起動すること
  - ファイアウォールに関する問題が出てきたら、再起動しないとダメな模様



# とりあえず書いてみよう



## クライアント (Client1.txt)

```
import processing.net.*;
Server myServer
    = new Server(this, 12345);
int count = 0;

void setup() {
    size(400, 200);
}

void draw() {
}

void stop(){
    myServer.stop();
}



void mousePressed() {
    count++;
    myServer.write(count);
}
```

## サーバ (Server1.txt)

```
import processing.net.*;
Client myClient = new Client( this,
    "127.0.0.1", 12345 );
int recvInt;
void setup() {
    size(400, 200);
    textSize( 200 );
    fill( 0 );
}
void stop(){
    myClient.stop();
}

void draw() {
    background( 255 );
    if (myClient.available() > 0) {
        recvInt = myClient.read();
    }
    text( recvInt, 100, 200 );
}
```

127.0.0.1 は  
自分という意味



```
sketch_131113a | Processing 2.0.3
File Edit Sketch Tools Help
sketch_131113a
import processing.net.*;
Server myServer;
int count = 0;

void setup() {
  size(400, 200);
  myServer = new Server(this, 12345);
}

void mousePressed() {
```

サーバ (Server)

```
sketch_131113b | Processing 2.0.3
File Edit Sketch Tools Help
sketch_131113b
Client myClient;
int dataIn;

void setup() {
  size(400, 200);
  textSize(200);
  myClient = new Client(this,
}

void draw() {
  background(255);
```

クライアント (Client)

サーバ (Server) から先に起動すること！  
( 窓口が準備できていないとだめ！ )


サーバ側でクリック連打してみよう！！

# とはいえ



- なんだか、自分の中でServerとかClientとか言ってもよくわからない。
- 隣の人とペアを作って接続してみよう！
  - 隣の人がない場合は、前後の余っている人同士でやってみる！
  - 一方はサーバ役，他方はクライアント役
    - 交互にやりましょう
  - まずは、隣の人アドレス（住所）を聞く！
    - 実際の住所ではなく、ネットワーク上の住所

# 隣に接続してみよう



# 隣に接続してみよう



```
C:\Windows\system32\cmd.exe
Microsoft Windows [Version 6.1.7601]
Copyright (c) 2009 Microsoft Corporation. All rights reserved.

C:\Users¥130007>
```

**ipconfig と入力して実行**

```
C:\Windows\system32\cmd.exe
Microsoft Windows [Version 6.1.7601]
Copyright (c) 2009 Microsoft Corporation. All rights reserved.

C:\Users¥130007>ipconfig
```

```
C:\Users¥130007>ipconfig

Windows IP 構成

イーサネット アダプター ローカル エリア接続:

    接続固有の DNS サフィックス . . . . . :
    IPv4 アドレス . . . . . : 133.26.54.5
    サブネット マスク . . . . . : 255.255.254.0
    デフォルト ゲートウェイ . . . . . : 133.26.54.254
```

**IPv4アドレスが各自の住所**

**IPv4アドレスを隣の人に伝えよう**

# とりあえず書いてみよう



```
import processing.net.*;
Client myClient = new Client( this, "133.26.54.5", 12345 );
int dataIn;

void setup() {
  size(400, 200);
  textSize( 200 );
  fill( 0 );
}
void stop(){
  myClient.stop();
}
void draw() {
  background( 255 );
  if (myClient.available() > 0) {
    dataIn = myClient.read();
  }
  text( dataIn, 100, 200 );
}
```

Server役の人の  
IPアドレスが133.26.54.5の場合

クライアント ( Client )

# 準備と相互接続



- サーバ・クライアント用のライブラリの準備

```
import processing.net.*;
```

- サーバの準備 + 接続待ち

```
Server myServer = new Server( this,  
                             接続待ちポート番号 );
```

- クライアントの準備 + 接続

```
Client myClient = new Client( this,  
                              "接続先アドレス", 接続先ポート番号 );
```

# メッセージの送信



- サーバからクライアントへのメッセージの送信（整数または文字列を送信）

```
myServer.write( 100 );  
myServer.write( "hoge" );
```

- クライアントからサーバへのメッセージの送信（整数または文字列を送信）

```
myClient.write( 100 );  
myClient.write( "hoge" );
```



# クライアントでのデータ受信



- クライアントでのデータ受信（整数）

```
int recvInt = myClient.read();
```

- クライアントでのデータ受信（文字列）

```
String recvStr = myClient.readString();
```

- クライアントでのデータ受信量取得

```
int size = myClient.available();
```

# サーバでのデータ受信



- データ受信待ちしているクライアントの取得

```
Client nextClient = myServer.available();
```

- あるクライアントからのデータ受信（整数）

```
int recvInt = nextClient.read();
```

- あるクライアントからのデータ受信（文字列）

```
String recvStr = nextClient.readString();
```

# 通信終了



- サーバからの切断

```
myClient.stop();
```

- 特定のクライアントの切断

```
myServer.disconnect( thisClient );
```

- サーバのサービスの終了

```
myServer.stop();
```

# 解説



133.26.54.5 さん

```
import processing.net.*;
Server myServer
    = new Server(this, 12345);
int count = 0;

void setup() {
    size(400, 200);
}

void draw() {
}

void stop(){
    myServer.stop();
}

void mousePressed() {
    count++;
    myServer.write(count);
}
```

12345番ポートで準備

クリック時にcountを増やし  
クライアントにcountを送信！

133.26.54.6 さん

```
import processing.net.*;
Client myClient = new Client( this,
    "133.26.54.5", 12345 );
int recvInt;

void setup() {
    size(400, 200);
    textSize( 200 );
    fill( 0 );
}

void stop(){
    myClient.stop();
}

void draw() {
    background( 255 );
    if (myClient.available() > 0) {
        recvInt = myClient.read();
    }
    text( recvInt, 100, 200 );
}
```

133.26.54.5さんの  
12345番ポートに接続

available() で受信  
しているかチェック  
0より大きければ受信

read() で値を受信して表示！

# 送受信の時に . . .



- どのようなフォーマット（形式）でデータを送受信するかを考える
- クリック数は1つの数字だけだが，ある座標を送受信しようと思うと，2つの値が必要となるため困ったことに

```
myServer.write( mouseX );  
myServer.write( mouseY );
```



```
myServer.write( mouseX + ", " + mouseY );
```

# 送受信の時に




ファイル入出力でもやったよね！

1. カンマ区切りで文字列として送信（他でもOK）
  - "x座標,y座標" （例） 100,230
  - XXXX.**write( mouseX+", "+mouseY );**
2. 文字列として受信
  - String recvStr = YYYY.**readString();**
3. 受信した文字列（カンマ区切り）を, splitを利用してカンマで分割  
（"100,230" → "100" と "230" に）
  - String [] data = **split( recvStr, ',' );**
4. 分割された文字列を整数に戻す
  - int x = **int( data[0] );**
  - int y = **int( data[1] );**

# クライアントに描画する



- サーバで操作することで、クライアント上に描画してみる
  - サーバとクライアントで相互接続
  - サーバ上でクリックされた位置をクライアントに送信
  - クライアントで送信されてきた座標に円を描画



# サーバから描画してみる

明治大学総合数理学部  
先端メディアサイエンス学科  
中村研究室



## サーバ ( Server2.txt )

```
// ライブラリを読み込む
import processing.net.*;
// サーバを2222ポートでスタート
Server myServer = new Server( this, 2222 );

void setup() {
  size(400, 200);
  noFill();
}

void mousePressed() {
  // クリック座標をカンマ区切りで送信
  myServer.write(mouseX+", "+mouseY);
}

void draw(){
}

void stop(){
  myServer.stop();
}
```

## クライアント ( Client2.txt )

```
// ライブラリを読み込む
import processing.net.*;
// X.X.X.X の 2222 番に接続
Client myClient = new Client(this, "X.X.X.X", 2222);
void setup() {
  size(400, 200);
  noFill();
  background( 255 );
}

void stop(){
  myClient.stop();
}



void draw() {
  // 何か受信しているか確認
  if (myClient.available() > 0) {
    // 文字列として受信
    String msg = myClient.readString();
    // 文字列をカンマ区切りで分割し整数に戻す
    String [] data = split( msg, ',' );
    ellipse( int(data[0]), int(data[1]), 50, 50 );
  }
}
```



# クライアントに描画してみる



- 一方向だとなんだか面白く無い
- 双方向にしてみよう！
  - サーバでクリックするとクライアントに，クライアントでクリックするとサーバに！



sketch\_131114a | Processing 2.0.3

File Sketch Tools Help



Java

sketch\_131114c | Processing 2.0.3

File Edit Sketch Tools Help



# 双方向で描画してみる



## サーバ (Server)

```
import processing.net.*;
Server myServer = new Server( this, 2222 );
void setup() {
  size(400, 200);
  noFill();
}
void mousePressed() {
  myServer.write(mouseX+", "+mouseY);
}
void stop(){
  myServer.stop();
}
void draw(){
  // データ送信しているクライアントの確認
  Client nextClient = myServer.available();
  // != null はクライアントがあるという意味
  if( nextClient != null ){
    // readString() でデータ受信
    String recvStr = nextClient.readString();
    String [] data = split( recvStr, ',' );
    ellipse( int(data[0]), int(data[1]), 50, 50 );
  }
}
```

## クライアント (Client)

```
import processing.net.*;
Client myClient = new Client(this, "X.X.X.X", 2222);

void setup() {
  size(400, 200);
  noFill();
  background( 255 );
}

void mousePressed(){
  myClient.write( mouseX+", "+mouseY );
}

void stop(){
  myClient.stop();
}

void draw() {
  if (myClient.available() > 0) {
    String recvStr = myClient.readString();
    String [] data = split( recvStr, ',' );
    ellipse( int(data[0]), int(data[1]), 50, 50 );
  }
}
```

# 双方向で描画してみる v.2



## サーバ (Server)

```
import processing.net.*;
Server myServer = new Server( this, 2222 );
void setup() {
  size(400, 200);
  noFill();
}

void mousePressed() {
  myServer.write(mouseX+", "+mouseY);
  stroke( 255, 0, 0 );
  ellipse( mouseX, mouseY, 50, 50 );
}

void stop(){
  myServer.stop();
}

void draw(){
  Client nextClient = myServer.available();
  if( nextClient != null ){
    String recvStr = nextClient.readString();
    String [] data = split( recvStr, ',' );
    stroke( 0, 0, 255 );
    ellipse( int(data[0]), int(data[1]), 50, 50);
  }
}
```

## クライアント (Client)

```
import processing.net.*;
Client myClient = new Client(this, "X.X.X.X", 2222);
void setup() {
  size(400, 200);
  noFill();
  background( 255 );
}

void mousePressed(){
  myClient.write( mouseX+", "+mouseY );
  stroke( 255, 0, 0 );
  ellipse( mouseX, mouseY, 50, 50 );
}

void stop(){
  myClient.stop();
}

void draw() {
  if (myClient.available() > 0) {
    String recvStr = myClient.readString();
    String [] data = split( recvStr, ',' );
    stroke( 0, 0, 255 );
    ellipse( int(data[0]), int(data[1]), 50, 50 );
  }
}
```

# クライアントから命令！



- サーバ役とクライアント役，どちらかがどちらかになって下さい．また，IPアドレスをそれぞれサーバ役の人のにし，サーバを実行し，次にクライアントを実行してみよう！

# サーバ (TrojanServer.txt)

明治大学総合数理学部  
先端メディアサイエンス学科  
中村研究室



```
import processing.net.*;
import java.awt.*;
import java.awt.event.*;
Server myServer = new Server( this, 3333 );
Robot robot;
```

**3333番ポートで準備**

```
void setup() {
  size(20, 20);
  try {
    robot = new Robot();
  } catch ( Exception e ) {
  }
}
```

```
void stop(){
  myServer.stop();
}
```

```
void draw() {
  Client nextClient = myServer.available();
  if ( nextClient != null ) {
    String recvStr = nextClient.readString();
    String [] data = split( recvStr, ',' );
    if ( data[0].equals( "move" ) == true ) {
      doMove( int( data[1] ), int( data[2] ) );
    }
  }
}
```

**最初のカンマの前が  
動作に関すること**

```
void doClick( int x, int y ) {
  robot.mousePress( InputEvent.BUTTON1_MASK );
  delay(1);
  robot.mouseRelease( InputEvent.BUTTON1_MASK );
  delay(1);
}
```

```
void doMove( int x, int y ) {
  robot.mouseMove( x, y );
}
```

# クライアント (TrojanClient.txt)



```
import processing.net.*;
Client myClient = new Client( this, "133.26.54.5", 3333 );

void setup() {
  size( 1200, 800 );
}
void stop(){
  myClient.stop();
}
void draw() {
}
// マウスが押された時
void mousePressed() {
  // "click,100,80," のように送信
}
// マウスが移動した時
void mouseMoved() {
  // "move,100,80," のように送信
  myClient.write( "move,"+mouseX+", "+mouseY+", " );
}
```

133.26.54.5さんの  
3333番ポートに接続

write() で  
メッセージを送信

# やってること



- サーバのマウスを操作してみよう！
  - “move, x, y” でマウスの移動情報を送信
- トロイの木馬でやっているのは，これの更に高度なこと



**演習： クリックを追加してみよう！**

# 参考：文字入力



- RobotのKeyPressを利用することで、相手のコンピュータに対して文字入力も指示できる！
- 興味がある人はやってみよう！！
- ライブラリのロード
  - `import java.awt.event.KeyEvent;`
- キープレス
  - `robot.KeyPress( KeyEvent.VK_キーの種類 );`





- RobotのcreateScreenCaptureを利用することで、画面のスクリーンキャプチャを撮ることが可能
- つまり、相手のデスクトップを見ることが可能
  - リモートからコンピュータを操作する際には、こういった組み合わせでやっています

# LINEもどきを作ってみよう

明治大学総合数理学部  
先端メディアサイエンス学科  
中村研究室



- スタンプを相手に貼り付けよう！



- LINEもどきを作ろうをダウンロードして改良しよう！

# サーバに情報を送ろう！



- 中央サーバをクリックの度にサーバに対してメッセージを送り続けるクライアントプログラムを作成せよ
  - write( "click" ) で文字列データを送信せよ
  - ポート番号は 54321 とする

**サーバの負荷を減らすのは大変**

# サーバに情報を送ろう！

明治大学総合数理学部  
先端メディアサイエンス学科  
中村研究室



```
import processing.net.*;
Client myClient = new Client(this, ,  );
int click = 0;
void setup() {
  size( 100, 100 );
}
void stop(){
  myClient.stop();
}
void draw(){
}


void mousePressed() {
  myClient.write(  );
}
```

# 簡易ウェブサーバを作ろう

明治大学総合数理学部  
先端メディアサイエンス学科  
中村研究室



- Webサーバってどんな仕組みで成り立っているのだろうか？（詳細はWebを支える技術参



<http://localhost:1000/index.html>



# 簡易Webサーバを作ろう(1)



```
// ライブラリを読み込む
import processing.net.*;
// サーバを1000ポートでスタート
Server myServer = new Server( this, 1000 );

void draw() {
  Client nextClient = myServer.available();
  if ( nextClient != null ) {
    println( nextClient.readString() );
    nextClient.write( "HTTP/1.1 200 OK\r\n" );
    nextClient.write( "Server: myServer\r\n" );
    nextClient.write( "Content-type: text/html\r\n" );
    nextClient.write( "\r\n" );
    nextClient.write( "Hello World" );
    myServer.disconnect( nextClient );
  }
}

void stop() {
  myServer.stop();
}
```

**writeでHTMLのヘッダと  
Hello Worldを返す！**

**最後にdisconnectで切断**





- 簡易Webサーバへのアクセス方法
  - ブラウザのURLに <http://localhost:1000/> と入力してみよう！
  - 隣の人サーバにアクセスしてみよう  
(localhostをIPに変更するだけ)
- リクエストに応じて処理を変えてみよう
  - <http://localhost:1000/hello.html>
  - <http://localhost:1000/fms.html>
- ブラウザはどんなリクエストを送ってきているのだろうか？

# 繋がる電光掲示板を作ろう



- 20x10の電光掲示板を作成し，サーバまたはクライアントの操作で，サーバとクライアントが同期して色が変わるようにするにはどうするか？



座標データを送り合えば良い？



# 参考：フォーマット



- データを送りまくるプログラムを作るとそのままではうまくいかなくなる
  - 例えば、下記のように座標情報を送信

100,100

110,110

110,120

120,130

## 参考：ぎなた読み

この先生きのこるには  
弁慶がなぎなたをもって  
スク水揚げ  
中国船さんご密漁

100,100 110,110 110,120 120,130

- 受信する場合は、どのようになるかは不明

100,100110,1

10110,120120,130

# 参考：フォーマット



- どこでデータが切れているのかという情報を付与する必要あり

[100, 100]	100, 100¥n	¥n は改行
[110, 110]	110, 110¥n	
[110, 120]	110, 120¥n	
[120, 130]	120, 130¥n	

- 受信したものを適切に処理する

[100, 100][110, 1	100, 100¥n110, 1
10][110, 120][120, 130]	10¥n110, 120¥n120, 130¥n

- 受信したものをバッファに追加し、後処理する

# 送信と受信をしっかりと



```
import processing.net.*;
Server myServer = new Server( this, 2222 );
String strBuffer;

void setup() {
    size(400, 200);
}
void stop(){
    myServer.stop();
}
void draw(){
    Client nextClient = myServer.available();
    if( nextClient != null ){
        String recvStr = nextClient.readString();
        strBuffer += recvStr;
    }
    // 改行までを処理する
    int pos = strBuffer.indexOf("\n");
    while( pos != -1 ){
        // 改行の手前までを取得する
        String strTarget = strBuffer.substring( 0, pos );
        String [] m = split( strTarget, "," );
        // m[0] と m[1] で処理する
        // \n の1文字後から次のバッファに放り込む
        strBuffer = strBuffer.substring( pos+1 );
        int pos = strBuffer.indexOf( "\n" );
    }
}
```

複数のクライアントが  
接続してきている場合は  
nextClient.ip();  
で、接続IPを取得し、  
それを利用して処理しよう！

# 陣取りゲームを作ってみよう


明治大学総合数理学部  
先端メディアサイエンス学科  
中村研究室



- 100x100のマス目のオセロのような陣取りゲーム
- クライアント上には100x100のマス目を表示
- マウスクリックした場所を, サーバ ( 教卓のIPアドレス ) に接続し, サーバに対してグループ番号と, マス目のx, y座標 ( それぞれ0~99 ) を送信する
- ポート番号 6666, 送信フォーマット [5,52,25] のように []で囲いカンマ区切りとする

番号

座標





```
import processing.net.*;
Client myClient = new Client( this,  ,  );
int unitSize = 8;
int ux = 100;
int uy = 100;
int [][] unit;
int myID =  ;

void setup() {
  size( 800, 800 );
  stroke( 0 );
  fill( 255 );
}

void draw() {
  for ( int x=0; x<ux; x++ ) {
    for ( int y=0; y<uy; y++ ) {
      rect( x*unitSize, y*unitSize, unitSize, unitSize );
    }
  }
}

void mousePressed() {
  int x = mouseX / unitSize;
  int y = mouseY / unitSize;
  myClient.write(  );
}
```

## 100x100のクライアント

# 課題1



- 講義資料に登場したLINEもどきを提出せよ
  - プログラム名はLineServer, LineClient
- 画像や操作をオリジナルのものから変更し, 工夫してみよう


# 課題2 questionnaireSC



クライアント (questionnaireClient) で回答するアンケートを作成する。クライアントでは写真を提示し, 写真の下に1~5の5段階の回答を示すボタンを表示せよ。また, Nextボタンで回答結果をサーバに送信するようにせよ。サーバに送信するときには [X問目, 評価] という形で送信せよ。


サーバ (questionnaireServer) では単純にアンケート結果を標準出力せよ。グラフで徐々に表示されるよう工夫しても良い。研究室のメンバーにアンケートに答えてもらいましょう

好きな度合いを回答して下さい  
(1:かなり嫌い~5:かなり好き)



① ② ③ ④ ⑤

Next



[1, 5]

# 発表会（案）



- テーマ： FMSの1年次ゼミにまつわる何かをプログラミングで表現せよ
  - ただし, 2つ以上のゼミを混ぜても良い
  - できるだけ他のグループとかぶらないように, かぶったらプレゼンのときに切なくなりますので…
  - 発表順は, 与えられたグループ番号順
- 条件
  - グループの全員が1つ以上のクラスを作ること
  - ファイル入出力, ネットワーク, WebAPI, フィジカルコンピューティングのうち2つ以上の技術を使うこと
- 提出物（予定）
  - 企画書, プログラム, A4のプレゼンボード（去年から変わりました）, 利用の説明書